Chapter 67

The TTEST Procedure

Chapter Table of Contents

Chapter 67

The TTEST Procedure

Overview

The TTEST procedure performs t tests for one sample, two samples, and paired observations. The one-sample t test compares the mean of the sample to a given number. The two-sample t test compares the mean of the first sample minus the mean of the second sample to a given number. The paired observations t test compares the mean of the differences in the observations to a given number.

For one-sample tests, PROC TTEST computes the sample mean of the variable and compares it with a given number. Paired comparisons use the one sample process on the differences between the observations. Paired comparisons can be made between many pairs of variables with one call to PROC TTEST. For group comparisons, PROC TTEST computes sample means for each of two groups of observations and tests the hypothesis that the population means differ by a given amount. This latter analysis can be considered a special case of a one-way analysis of variance with two levels of classification.

The underlying assumption of the t test in all three cases is that the observations are random samples drawn from normally distributed populations. This assumption can be checked using the UNIVARIATE procedure; if the normality assumptions for the t test are not satisfied, you should analyze your data using the NPAR1WAY procedure. The two populations of a group comparison must also be independent. If they are not independent, you should question the validity of a paired comparison.

PROC TTEST computes the group comparison t statistic based on the assumption that the variances of the two groups are equal. It also computes an approximate t based on the assumption that the variances are unequal (the Behrens-Fisher problem). The degrees of freedom and probability level are given for each; Satterthwaite's (1946) approximation is used to compute the degrees of freedom associated with the approximate t. In addition, you can request the Cochran and Cox (1950) approximation of the

probability level for the approximate t. The folded form of the F statistic is computed to test for equality of the two variances (Steel and Torrie 1980).

FREQ and WEIGHT statements are available. Data can be input in the form of observations or summary statistics. Summary statistics and their confidence intervals, and differences of means are output. For two-sample tests, the pooled-variance and a test for equality of variances are also produced.

Getting Started

One-Sample *t* Test

A one-sample t test can be used to compare a sample mean to a given value. This example, taken from Huntsberger and Billingsley (1989, p. 290), tests whether the mean length of a certain type of court case is 80 days using 20 randomly chosen cases. The data are read by the following DATA step:

```
title 'One-Sample t Test';
data time;
   input time @@;
   datalines;
 43  90  84  87  116   95  86   99   93  92
121  71  66  98   79  102  60  112  105  98
;
run;
```

The only variable in the data set, time, is assumed to be normally distributed. The trailing at signs (@@) indicate that there is more than one observation on a line. The following code invokes PROC TTEST for a one-sample t test:

```
proc ttest h0=80 alpha=0.1;
   var time;
run;
```

The VAR statement indicates that the time variable is being studied, while the H0= option specifies that the mean of the time variable should be compared to the value 80 rather than the default null hypothesis of 0. This ALPHA= option requests 10% confidence intervals rather than the default 5% confidence intervals. The output is displayed in Figure 67.1.

One-Sample t Test

The TTEST Procedure

Statistics

Variable	N	Lower CL Mean	Mean	Upper CL Mean	Lower CL Std Dev	Std Dev	Upper CL Std Dev	Std Err	Minimum	Maximum
time	20	82.447	89.85	97.253	15.2	19.146	26.237	4.2811	43	121

T-Tests

Variable	DF	t Value	Pr > |t|
time	19	2.30	0.0329

Figure 67.1. One-Sample t Test Results

Summary statistics appear at the top of the output. The sample size (N), the mean and its confidence bounds (Lower CL Mean and Upper CL Mean), the standard deviation and its confidence bounds (Lower CL Std Dev and Upper CL Std Dev), and the standard error are displayed with the minimum and maximum values of the time variable. The test statistic, the degrees of freedom, and the p-value for the t test are displayed next; at the 10% α-level, this test indicates that the mean length of the court cases are significantly different from 80 days $(t = 2.30, p = 0.0329)$.

Comparing Group Means

If you want to compare values obtained from two different groups, and if the groups are independent of each other and the data are normally distributed in each group, then a group t test can be used. Examples of such group comparisons include

- test scores for two third-grade classes, where one of the classes receives tutoring
- fuel efficiency readings of two automobile nameplates, where each nameplate uses the same fuel
- sunburn scores for two sunblock lotions, each applied to a different group of people
- political attitude scores of males and females

In the following example, the golf scores for males and females in a physical education class are compared. The sample sizes from each population are equal, but this is not required for further analysis. The data are read by the following statements:

```
title 'Comparing Group Means';
data scores;
   input Gender $ Score @@;
   datalines;
f 75  f 76  f 80  f 77  f 80  f 77  f 73
m 82  m 80  m 85  m 85  m 78  m 87  m 82
;
run;
```

The dollar sign ($) following Gender in the INPUT statement indicates that Gender is a character variable. The trailing at signs (@@) enable the procedure to read more than one observation per line.

You can use a group t test to determine if the mean golf score for the men in the class differs significantly from the mean score for the women. If you also suspect that the distributions of the golf scores of males and females have unequal variances, then submitting the following statements invokes PROC TTEST with options to deal with the unequal variance case.

```
proc ttest cochran ci=equal umpu;
   class Gender;
   var Score;
run;
```

The CLASS statement contains the variable that distinguishes the groups being compared, and the VAR statement specifies the response variable to be used in calculations. The COCHRAN option produces p-values for the unequal variance situation using the Cochran and Cox(1950) approximation. Equal tailed and uniformly most powerful unbiased (UMPU) confidence intervals for σ are requested by the CI= option. Output from these statements is displayed in Figure 67.2 through Figure 67.4.

Comparing Group Means

The TTEST Procedure

Statistics

Variable	Class	N	Lower CL Mean	Mean	Upper CL Mean	Lower CL Std Dev	UMPU Lower CL Std Dev	Std Dev
Score	f	7	74.504	76.857	79.211	1.6399	1.5634	2.5448
Score	m	7	79.804	82.714	85.625	2.028	1.9335	3.1472
Score	Diff (1-2)		-9.19	-5.857	-2.524	2.0522	2.0019	2.8619

Statistics

Variable	Class	UMPU Upper CL Std Dev	Upper CL Std Dev	Std Err	Minimum	Maximum
Score	f	5.2219	5.6039	0.9619	73	80
Score	m	6.4579	6.9303	1.1895	78	87
Score	Diff (1-2)	4.5727	4.7242	1.5298		

Figure 67.2. Simple Statistics

Simple statistics for the two populations being compared, as well as for the difference of the means between the populations, are displayed in Figure 67.2. The Variable column denotes the response variable, while the Class column indicates the population corresponding to the statistics in that row. The sample size (N) for each population, the sample means (Mean), and lower and upper confidence bounds for the means (Lower CL Mean and Upper CL Mean) are displayed next. The standard deviations (Std Dev) are displayed as well, with equal tailed confidence bounds in the Lower CL Std Dev and Upper CL Std Dev columns and UMPU confidence bounds in the UMPU Upper CL Std Dev and UMPU Lower CL Std Dev columns. In addition, standard error of the mean and the minimum and maximum data values are displayed.

Comparing Group Means

The TTEST Procedure

T-Tests

Variable	Method	Variances	DF	t Value	Pr > |t|
Score	Pooled	Equal	12	-3.83	0.0024
Score	Satterthwaite	Unequal	11.5	-3.83	0.0026
Score	Cochran	Unequal	6	-3.83	0.0087

Figure 67.3. t Tests

The test statistics, associated degrees of freedom, and p-values are displayed in Figure 67.3. The Method column denotes which t test is being used for that row, and the Variances column indicates what assumption about variances is being made. The pooled test assumes that the two populations have equal variances and uses degrees of freedom $n_1 + n_2 - 2$, where n_1 and n_2 are the sample sizes for the two populations. The remaining two tests do not assume that the populations have equal variances. The Satterthwaite test uses the Satterthwaite approximation for degrees of freedom, while the Cochran test uses the Cochran and Cox approximation for the p-value.

Comparing Group Means

The TTEST Procedure

Equality of Variances

Variable	Method	Num DF	Den DF	F Value	Pr > F
Score	Folded F	6	6	1.53	0.6189

Figure 67.4. Tests of Equality of Variances

Examine the output in Figure 67.4 to determine which t test is appropriate. The “Equality of Variances” test results show that the assumption of equal variances is reasonable for these data (the Folded F statistic $F' = 1.53$, with $p = 0.6189$). If the assumption of normality is also reasonable, the appropriate test is the usual pooled t test, which shows that the average golf scores for men and women are significantly different $(t = -3.83, p = 0.0024)$. If the assumption of equality of variances is not reasonable, then either the Satterthwaite or the Cochran test should be used.

The assumption of normality can be checked using PROC UNIVARIATE; if the assumption of normality is not reasonable, you should analyze the data with the non-parametric Wilcoxon Rank Sum test using PROC NPAR1WAY.

Syntax

The following statements are available in PROC TTEST.

PROC TTEST < *options* > **;**
  CLASS *variable* **;**
  PAIRED *variables* **;**
  BY *variables* **;**
  VAR *variables* **;**
  FREQ *variable* **;**
  WEIGHT *variable* **;**

No statement can be used more than once. There is no restriction on the order of the statements after the PROC statement.

PROC TTEST Statement

PROC TTEST < *options* > **;**

The following options can appear in the PROC TTEST statement.

ALPHA=*p*
specifies that confidence intervals are to be $100(1-p)\%$ confidence intervals, where $0 < p < 1$. By default, PROC TTEST uses ALPHA=0.05. If p is 0 or less, or 1 or more, an error message is printed.

CI=EQUAL
CI=UMPU
CI=NONE
specifies whether a confidence interval is displayed for σ and, if so, what kind. The CI=EQUAL option specifies an equal tailed confidence interval, and it is the default. The CI=UMPU option specifies an interval based on the uniformly most powerful unbiased test of $H_0{:}\,\sigma = \sigma_0$. The CI=NONE option requests that no confidence interval be displayed for σ. The values EQUAL and UMPU together request that both types of confidence intervals be displayed. If the value NONE is specified with one or both of the values EQUAL and UMPU, NONE takes precedence. For more information, see the "Confidence Interval Estimation" section on page 3579.

COCHRAN
requests the Cochran and Cox (1950) approximation of the probability level of the approximate t statistic for the unequal variances situation.

DATA=*SAS-data-set*
names the SAS data set for the procedure to use. By default, PROC TTEST uses the most recently created SAS data set. The input data set can contain summary statistics of the observations instead of the observations themselves. The number, mean, and standard deviation of the observations are required for each BY group (one sample and paired differences) or for each class within each BY group (two samples). For

more information on the DATA= option, see the "Input Data Set of Statistics" section on page 3577.

H0=*m*

requests tests against m instead of 0 in all three situations (one-sample, two-sample, and paired observation t tests). By default, PROC TTEST uses H0=0.

BY Statement

BY *variables* **;**

You can specify a BY statement with PROC TTEST to obtain separate analyses on observations in groups defined by the BY variables. When a BY statement appears, the procedure expects the input data set to be sorted in order of the BY variables.

If your input data set is not sorted in ascending order, use one of the following alternatives:

- Sort the data using the SORT procedure with a similar BY statement.
- Specify the BY statement option NOTSORTED or DESCENDING in the BY statement for the TTEST procedure. The NOTSORTED option does not mean that the data are unsorted but rather that the data are arranged in groups (according to values of the BY variables) and that these groups are not necessarily in alphabetical or increasing numeric order.
- Create an index on the BY variables using the DATASETS procedure (in base SAS software).

For more information on the BY statement, refer to the discussion in *SAS Language Reference: Concepts*. For more information on the DATASETS procedure, refer to the *SAS Procedures Guide*.

CLASS Statement

CLASS *variable* **;**

A CLASS statement giving the name of the classification (or grouping) variable must accompany the PROC TTEST statement in the two independent sample cases. It should be omitted for the one sample or paired comparison situations. If it is used without the VAR statement, all numeric variables in the input data set (except those appearing in the CLASS, BY, FREQ, or WEIGHT statement) are included in the analysis.

The class variable must have two, and only two, levels. PROC TTEST divides the observations into the two groups for the t test using the levels of this variable. You can use either a numeric or a character variable in the CLASS statement.

Class levels are determined from the formatted values of the CLASS variable. Thus, you can use formats to define group levels. Refer to the discussions of the FORMAT procedure, the FORMAT statement, formats, and informats in *SAS Language Reference: Dictionary*.

FREQ Statement

FREQ *variable* **;**

The *variable* in the FREQ statement identifies a variable that contains the frequency of occurrence of each observation. PROC TTEST treats each observation as if it appears n times, where n is the value of the FREQ variable for the observation. If the value is not an integer, only the integer portion is used. If the frequency value is less than 1 or is missing, the observation is not used in the analysis. When the FREQ statement is not specified, each observation is assigned a frequency of 1. The FREQ statement cannot be used if the DATA= data set contains statistics instead of the original observations.

PAIRED Statement

PAIRED *PairLists* **;**

The *PairLists* in the PAIRED statement identifies the variables to be compared in paired comparisons. You can use one or more *PairLists*. Variables or lists of variables are separated by an asterisk (*) or a colon (:). The asterisk requests comparisons between each variable on the left with each variable on the right. The colon requests comparisons between the first variable on the left and the first on the right, the second on the left and the second on the right, and so forth. The number of variables on the left must equal the number on the right when the colon is used. The differences are calculated by taking the variable on the left minus the variable on the right for both the asterisk and colon. A pair formed by a variable with itself is ignored. Use the PAIRED statement only for paired comparisons. The CLASS and VAR statements cannot be used with the PAIRED statement.

Examples of the use of the asterisk and the colon are shown in the following table.

These PAIRED statements...	yield these comparisons
`PAIRED A*B;`	A-B
`PAIRED A*B C*D;`	A-B and C-D
`PAIRED (A B)*(C D);`	A-C, A-D, B-C, and B-D
`PAIRED (A B)*(C B);`	A-C, A-B, and B-C
`PAIRED (A1-A2)*(B1-B2);`	A1-B1, A1-B2, A2-B1, and A2-B2
`PAIRED (A1-A2):(B1-B2);`	A1-B1 and A2-B2

VAR Statement

VAR *variables* **;**

The VAR statement names the variables to be used in the analyses. One-sample comparisons are conducted when the VAR statement is used without the CLASS statement, while group comparisons are conducted when the VAR statement is used with a CLASS statement. If the VAR statement is omitted, all numeric variables in the input data set (except a numeric variable appearing in the BY, CLASS, FREQ, or WEIGHT statement) are included in the analysis. The VAR statement can be used with one- and two-sample t tests and cannot be used with the PAIRED statement.

WEIGHT Statement

WEIGHT *variable* **;**

The WEIGHT statement weights each observation in the input data set by the value of the WEIGHT variable. The values of the WEIGHT variable can be nonintegral, and they are not truncated. Observations with negative, zero, or missing values for the WEIGHT variable are not used in the analyses. Each observation is assigned a weight of 1 when the WEIGHT statement is not used. The WEIGHT statement cannot be used with an input data set of summary statistics.

Details

Input Data Set of Statistics

PROC TTEST accepts data containing either observation values or summary statistics. It assumes that the DATA= data set contains statistics if it contains a character variable with name _TYPE_ or _STAT_. The TTEST procedure expects this character variable to contain the names of statistics. If both _TYPE_ and _STAT_ variables exist and are of type character, PROC TTEST expects _TYPE_ to contain the names of statistics including ‘N’, ‘MEAN’, and ‘STD’ for each BY group (or for each class within each BY group for two-sample t tests). If no ‘N’, ‘MEAN’, or ‘STD’ statistics exist, an error message is printed.

FREQ, WEIGHT, and PAIRED statements cannot be used with input data sets of statistics. BY, CLASS, and VAR statements are the same regardless of data set type. For paired comparisons, see the _DIF_ values for the _TYPE_=T observations in output produced by the OUTSTATS= option in the PROC COMPARE statement (refer to the *SAS Procedures Guide*).

Missing Values

An observation is omitted from the calculations if it has a missing value for either the CLASS variable, a PAIRED variable, or the variable to be tested. If more than one variable is listed in the VAR statement, a missing value in one variable does not eliminate the observation from the analysis of other nonmissing variables.

Computational Methods

The t Statistic

The form of the t statistic used varies with the type of test being performed.

- To compare an individual mean with a sample of size n to a value m, use

  $$t = \frac{\bar{x} - m}{s/\sqrt{n}}$$

  where $\bar{x}$ is the sample mean of the observations and s^2 is the sample variance of the observations.

- To compare n paired differences to a value m, use

  $$t = \frac{\bar{d} - m}{s_d/\sqrt{n}}$$

  where $\bar{d}$ is the sample mean of the paired differences and s_d^2 is the sample variance of the paired differences.

- To compare means from two independent samples with n_1 and n_2 observations to a value m, use

  $$t = \frac{(\bar{x}_1 - \bar{x}_2) - m}{s\sqrt{\frac{1}{n_1} + \frac{1}{n_2}}}$$

  where s^2 is the pooled variance

  $$s^2 = \frac{(n_1 - 1)s_1^2 + (n_2 - 1)s_2^2}{n_1 + n_2 - 2}$$

  and s_1^2 and s_2^2 are the sample variances of the two groups. The use of this t statistic depends on the assumption that $\sigma_1^2 = \sigma_2^2$, where σ_1^2 and σ_2^2 are the population variances of the two groups.

The Folded Form F Statistic

The folded form of the F statistic, F', tests the hypothesis that the variances are equal, where

$$F' = \frac{\max(s_1^2, s_2^2)}{\min(s_1^2, s_2^2)}$$

A test of F' is a two-tailed F test because you do not specify which variance you expect to be larger. The p-value gives the probability of a greater F value under the null hypothesis that $\sigma_1^2 = \sigma_2^2$.

The Approximate t Statistic

Under the assumption of unequal variances, the approximate t statistic is computed as

$$t' = \frac{\bar{x}_1 - \bar{x}_2}{\sqrt{w_1 + w_2}}$$

where

$$w_1 = \frac{s_1^2}{n_1}, \quad w_2 = \frac{s_2^2}{n_2}$$

The Cochran and Cox Approximation

The Cochran and Cox (1950) approximation of the probability level of the approximate t statistic is the value of p such that

$$t' = \frac{w_1 t_1 + w_2 t_2}{w_1 + w_2}$$

where t_1 and t_2 are the critical values of the t distribution corresponding to a significance level of p and sample sizes of n_1 and n_2, respectively. The number of degrees of freedom is undefined when $n_1 \neq n_2$. In general, the Cochran and Cox test tends to be conservative (Lee and Gurland 1975).

Satterthwaite's Approximation

The formula for Satterthwaite's (1946) approximation for the degrees of freedom for the approximate t statistic is:

$$df = \frac{(w_1 + w_2)^2}{\left(\frac{w_1^2}{n_1 - 1} + \frac{w_2^2}{n_2 - 1}\right)}$$

Refer to Steel and Torrie (1980) or Freund, Littell, and Spector (1986) for more information.

Confidence Interval Estimation

The form of the confidence interval varies with the statistic for which it is computed. In the following confidence intervals involving means, $t_{1-\frac{\alpha}{2},n-1}$ is the $100(1-\frac{\alpha}{2})\%$ quantile of the t distribution with $n-1$ degrees of freedom. The confidence interval for

- an individual mean from a sample of size n compared to a value m is given by

  $$(\bar{x} - m) \pm t_{1-\frac{\alpha}{2},n-1} \frac{s}{\sqrt{n}}$$

  where $\bar{x}$ is the sample mean of the observations and s^2 is the sample variance of the observations

- paired differences with a sample of size n differences compared to a value m is given by

  $$(\bar{d} - m) \pm t_{1-\frac{\alpha}{2},n-1} \frac{s_d}{\sqrt{n}}$$

  where $\bar{d}$ and s_d^2 are the sample mean and sample variance of the paired differences, respectively

- the difference of two means from independent samples with n_1 and n_2 observations compared to a value m is given by

  $$((\bar{x}_1 - \bar{x}_2) - m) \pm t_{1-\frac{\alpha}{2},n_1+n_2-2}\, s \sqrt{\frac{1}{n_1} + \frac{1}{n_2}}$$

  where s^2 is the pooled variance

  $$s^2 = \frac{(n_1 - 1)s_1^2 + (n_2 - 1)s_2^2}{n_1 + n_2 - 2}$$

  and where s_1^2 and s_2^2 are the sample variances of the two groups. The use of this confidence interval depends on the assumption that $\sigma_1^2 = \sigma_2^2$, where σ_1^2 and σ_2^2 are the population variances of the two groups.

The distribution of the estimated standard deviation of a mean is not symmetric, so alternative methods of estimating confidence intervals are possible. PROC TTEST computes two estimates. For both methods, the data are assumed to have a normal distribution with mean μ and variance σ^2, both unknown. The methods are as follows:

- The default method, an equal-tails confidence interval, puts an equal amount of area ($\frac{\alpha}{2}$) in each tail of the chi-square distribution. An equal tails test of $H_0 \colon \sigma = \sigma_0$ has acceptance region

  $$\left\{ \chi^2_{\frac{\alpha}{2},n-1} \leq \frac{(n-1)S^2}{\sigma_0^2} \leq \chi^2_{\frac{1-\alpha}{2},n-1} \right\}$$

  which can be algebraically manipulated to give the following $100(1 - \alpha)\%$ confidence interval for σ^2:

  $$\left(\frac{(n-1)S^2}{\chi^2_{1-\frac{\alpha}{2},n-1}}, \frac{(n-1)S^2}{\chi^2_{\frac{\alpha}{2},n-1}} \right)$$

  In order to obtain a confidence interval for σ, the square root of each side is taken, leading to the following $100(1 - \alpha)\%$ confidence interval:

  $$\left(\sqrt{\frac{(n-1)S^2}{\chi^2_{1-\frac{\alpha}{2},n-1}}}, \sqrt{\frac{(n-1)S^2}{\chi^2_{\frac{\alpha}{2},n-1}}} \right)$$

- The second method yields a confidence interval derived from the uniformly most powerful unbiased test of $H_0\colon \sigma = \sigma_0$ (Lehmann 1986). This test has acceptance region

  $$\left\{ c_1 \le \frac{(n-1)S^2}{\sigma_0^2} \le c_2 \right\}$$

  where the critical values c_1 and c_2 satisfy

  $$\int_{c_1}^{c_2} f_n(y)dy = 1-\alpha$$

  and

  $$\int_{c_1}^{c_2} y f_n(y)dy = n(1-\alpha)$$

  where $f_n(y)$ is the chi-squared distribution with n degrees of freedom. This acceptance region can be algebraically manipulated to arrive at

  $$P\left\{ \frac{(n-1)S^2}{c_2} \le \sigma^2 \le \frac{(n-1)S^2}{c_1} \right\} = 1-\alpha$$

  where c_1 and c_2 solve the preceding two integrals. To find the area in each tail of the chi-square distribution to which these two critical values correspond, solve $c_1 = \chi^2_{1-\alpha_2,n-1}$ and $c_2 = \chi^2_{\alpha_1,n-1}$ for α_1 and α_2; the resulting α_1 and α_2 sum to α. Hence, a $100(1-\alpha)\%$ confidence interval for σ^2 is given by

  $$\left(\frac{(n-1)S^2}{\chi^2_{1-\alpha_2,n-1}}, \frac{(n-1)S^2}{\chi^2_{\alpha_1,n-1}} \right)$$

  In order to obtain a $100(1-\alpha)\%$ confidence interval for σ, the square root is taken of both terms, yielding

  $$\left(\sqrt{\frac{(n-1)S^2}{\chi^2_{1-\alpha_2,n-1}}}, \sqrt{\frac{(n-1)S^2}{\chi^2_{\alpha_1,n-1}}} \right)$$

Displayed Output

For each variable in the analysis, the TTEST procedure displays the following summary statistics for each group:

- the name of the dependent variable
- the levels of the classification variable
- N, the number of nonmissing values
- Lower CL Mean, the lower confidence bound for the mean
- the Mean or average
- Upper CL Mean, the upper confidence bound for the mean
- Lower CL Std Dev, the lower confidence bound for the standard deviation
- Std Dev, the standard deviation
- Upper CL Std Dev, the upper confidence bound for the standard deviation
- Std Err, the standard error of the mean
- the Minimum value, if the line size allows
- the Maximum value, if the line size allows
- upper and lower UMPU confidence bounds for the standard deviation, displayed if the CI=UMPU option is specified in the PROC TTEST statement

Next, the results of several t tests are given. For one-sample and paired observations t tests, the TTEST procedure displays

- t Value, the t statistic for testing the null hypothesis that the mean of the group is zero
- DF, the degrees of freedom
- Pr > |t|, the probability of a greater absolute value of t under the null hypothesis. This is the two-tailed significance probability. For a one-tailed test, halve this probability.

For two-sample t tests, the TTEST procedure displays all the items in the following list. You need to decide whether equal or unequal variances are appropriate for your data.

- Under the assumption of unequal variances, the TTEST procedure displays results using Satterthwaite's method. If the COCHRAN option is specified, the results for the Cochran and Cox approximation are also displayed.
  - t Value, an approximate t statistic for testing the null hypothesis that the means of the two groups are equal
  - DF, the approximate degrees of freedom
  - Pr > |t|, the probability of a greater absolute value of t under the null hypothesis. This is the two-tailed significance probability. For a one-tailed test, halve this probability.

- Under the assumption of equal variances, the TTEST procedure displays results obtained by pooling the group variances.
  - t Value, the t statistic for testing the null hypothesis that the means of the two groups are equal
  - DF, the degrees of freedom
  - Pr > |t|, the probability of a greater absolute value of t under the null hypothesis. This is the two-tailed significance probability. For a one-tailed test, halve this probability.
- PROC TTEST then gives the results of the test of equality of variances:
  - the F' (folded) statistic (see the "The Folded Form F Statistic" section on page 3578)
  - Num DF and Den DF, the numerator and denominator degrees of freedom in each group
  - Pr > F, the probability of a greater F' value. This is the two-tailed significance probability.

ODS Table Names

PROC TTEST assigns a name to each table it creates. You can use these names to reference the table when using the Output Delivery System (ODS) to select tables and create output data sets. These names are listed in the following table. For more information on ODS, see Chapter 15, "Using the Output Delivery System."

Table 67.1. ODS Tables Produced in PROC TTEST

ODS Table Name	Description	Statement
Equality	Tests for equality of variance	CLASS statement
Statistics	Univariate summary statistics	by default
TTests	t-tests	by default

Examples

Example 67.1. Comparing Group Means Using Input Data Set of Summary Statistics

The following example, taken from Huntsberger and Billingsley (1989), compares two grazing methods using 32 steer. Half of the steer are allowed to graze continuously while the other half are subjected to controlled grazing time. The researchers want to know if these two grazing methods impact weight gain differently. The data are read by the following DATA step.

```
title 'Group Comparison Using Input Data Set of Summary
       Statistics';
data graze;
   length GrazeType $ 10;
   input GrazeType $ WtGain @@;
   datalines;
controlled  45   controlled  62
controlled  96   controlled 128
controlled 120   controlled  99
controlled  28   controlled  50
controlled 109   controlled 115
controlled  39   controlled  96
controlled  87   controlled 100
controlled  76   controlled  80
continuous  94   continuous  12
continuous  26   continuous  89
continuous  88   continuous  96
continuous  85   continuous 130
continuous  75   continuous  54
continuous 112   continuous  69
continuous 104   continuous  95
continuous  53   continuous  21
;
run;
```

The variable GrazeType denotes the grazing method: 'controlled' is controlled grazing and 'continuous' is continuous grazing. The dollar sign ($) following GrazeType makes it a character variable, and the trailing at signs (@@) tell the procedure that there is more than one observation per line. The MEANS procedure is invoked to create a data set of summary statistics with the following statements:

```
proc sort;
   by GrazeType;
proc means data=graze noprint;
   var WtGain;
   by GrazeType;
   output out=newgraze;
run;
```

The NOPRINT option eliminates all output from the MEANS procedure. The VAR statement tells PROC MEANS to compute summary statistics for the WtGain variable, and the BY statement requests a separate set of summary statistics for each level of GrazeType. The OUTPUT OUT= statement tells PROC MEANS to put the summary statistics into a data set called newgraze so that it may be used in subsequent procedures. This new data set is displayed in Output 67.1.1 by using PROC PRINT as follows:

```
proc print data=newgraze;
run;
```

The _STAT_ variable contains the names of the statistics, and the GrazeType variable indicates which group the statistic is from.

Output 67.1.1. Output Data Set of Summary Statistics

Group Comparison Using Input Data Set of Summary Statistics

Obs	GrazeType	_TYPE_	_FREQ_	_STAT_	WtGain
1	continuous	0	16	N	16.000
2	continuous	0	16	MIN	12.000
3	continuous	0	16	MAX	130.000
4	continuous	0	16	MEAN	75.188
5	continuous	0	16	STD	33.812
6	controlled	0	16	N	16.000
7	controlled	0	16	MIN	28.000
8	controlled	0	16	MAX	128.000
9	controlled	0	16	MEAN	83.125
10	controlled	0	16	STD	30.535

The following code invokes PROC TTEST using the newgraze data set, as denoted by the DATA= option.

```
proc ttest data=newgraze;
   class GrazeType;
   var WtGain;
run;
```

The CLASS statement contains the variable that distinguishes between the groups being compared, in this case GrazeType. The summary statistics and confidence intervals are displayed first, as shown in Output 67.1.2.

Output 67.1.2. Summary Statistics

Group Comparison Using Input Data Set of Summary Statistics

The TTEST Procedure

Statistics

Variable	Class	N	Lower CL Mean	Mean	Upper CL Mean	Lower CL Std Dev	Std Dev	Upper CL Std Dev	Std Err	Minimum	Maximum
WtGain	continuous	16	57.171	75.188	93.204	.	33.812	.	8.4529	12	130
WtGain	controlled	16	66.854	83.125	99.396	.	30.535	.	7.6337	28	128
WtGain	Diff (1-2)		-31.2	-7.938	15.323	25.743	32.215	43.061	11.39		

In Output 67.1.2, the Variable column states the variable used in computations and the Class column specifies the group for which the statistics are computed. For each class, the sample size, mean, standard deviation and standard error, and maximum and minimum values are displayed. The confidence bounds for the mean are also displayed; however, since summary statistics are used as input, the confidence bounds for the standard deviation of the groups are not calculated.

Output 67.1.3. t Tests

Group Comparison Using Input Data Set of Summary Statistics

The TTEST Procedure

T-Tests

Variable	Method	Variances	DF	t Value	Pr > |t|
WtGain	Pooled	Equal	30	-0.70	0.4912
WtGain	Satterthwaite	Unequal	29.7	-0.70	0.4913

Equality of Variances

Variable	Method	Num DF	Den DF	F Value	Pr > F
WtGain	Folded F	15	15	1.23	0.6981

Output 67.1.3 shows the results of tests for equal group means and equal variances. A group test statistic for the equality of means is reported for equal and unequal variances. Before deciding which test is appropriate, you should look at the test for equality of variances; this test does not indicate a significant difference in the two variances $(F' = 1.23, p = 0.6981)$, so the pooled t statistic should be used. Based on the pooled statistic, the two grazing methods are not significantly different $(t = 0.70, p = 0.4912)$. Note that this test assumes that the observations in both data sets are normally distributed; this assumption can be checked in PROC UNIVARIATE using the raw data.

Example 67.2. One-Sample Comparison Using the FREQ Statement

This example examines children's reading skills. The data consist of Degree of Reading Power (DRP) test scores from 44 third-grade children and are taken from Moore (1995, p. 337). Their scores are given in the following DATA step.

```
title 'One-Mean Comparison Using FREQ Statement';
data read;
   input score count @@;
   datalines;
40 2   47 2   52 2   26 1   19 2
25 2   35 4   39 1   26 1   48 1
14 2   22 1   42 1   34 2   33 2
18 1   15 1   29 1   41 2   44 1
51 1   43 1   27 2   46 2   28 1
49 1   31 1   28 1   54 1   45 1
;
run;
```

The following statements invoke the TTEST procedure to test if the mean test score is equal to 30. The count variable contains the frequency of occurrence of each test score; this is specified in the FREQ statement.

```
proc ttest data=read h0=30;
   var score;
   freq count;
run;
```

The output, shown in Output 67.2.1, contains the results.

Output 67.2.1. TTEST Results

```
                           One-Mean Comparison Using FREQ Statement

                                    The TTEST Procedure

Frequency: count


                                        Statistics

                  Lower CL            Upper CL  Lower CL           Upper CL
Variable       N      Mean      Mean      Mean   Std Dev   Std Dev   Std Dev   Std Err   Minimum   Maximum

score         44    31.449    34.864    38.278    9.2788     11.23    14.229     1.693        14        54


                                          T-Tests

                            Variable       DF    t Value    Pr > |t|

                            score          43       2.87      0.0063
```

The SAS log states that 30 observations and two variables have been read. However, the sample size given in the TTEST output is N=44. This is due to specifying the count variable in the FREQ statement. The test is significant $(t = 2.87, p = 0.0063)$ at the 5% level, thus you can conclude that the mean test score is different from 30.

Example 67.3. Paired Comparisons

When it is not feasible to assume that two groups of data are independent, and a natural pairing of the data exists, it is advantageous to use an analysis that takes the correlation into account. Utilizing this correlation results in higher power to detect existing differences between the means. The differences between paired observations are assumed to be normally distributed. Some examples of this natural pairing are

- pre- and post-test scores for a student receiving tutoring
- fuel efficiency readings of two fuel types observed on the same automobile
- sunburn scores for two sunblock lotions, one applied to the individual's right arm, one to the left arm
- political attitude scores of husbands and wives

In this example, taken from *SUGI Supplemental Library User's Guide, Version 5 Edition*, a stimulus is being examined to determine its effect on systolic blood pressure. Twelve men participate in the study. Their systolic blood pressure is measured both before and after the stimulus is applied. The following statements input the data:

```
title 'Paired Comparison';
data pressure;
   input SBPbefore SBPafter @@;
   datalines;
120 128   124 131   130 131   118 127
140 132   128 125   140 141   135 137
126 118   130 132   126 129   127 135
;
run;
```

The variables SBPbefore and SBPafter denote the systolic blood pressure before and after the stimulus, respectively.

The statements to perform the test follow.

```
proc ttest;
   paired SBPbefore*SBPafter;
run;
```

The PAIRED statement is used to test whether the mean change in systolic blood pressure is significantly different from zero. The output is displayed in Output 67.3.1.

Output 67.3.1. TTEST Results

Paired Comparison

The TTEST Procedure

Statistics

Difference	N	Lower CL Mean	Mean	Upper CL Mean	Lower CL Std Dev	Std Dev	Upper CL Std Dev	Std Err	Minimum	Maximum
SBPbefore - SBPafter	12	-5.536	-1.833	1.8698	4.1288	5.8284	9.8958	1.6825	-9	8

T-Tests

Difference	DF	t Value	Pr > |t|
SBPbefore - SBPafter	11	-1.09	0.2992

The variables SBPbefore and SBPafter are the paired variables with a sample size of 12. The summary statistics of the difference are displayed (mean, standard deviation, and standard error) along with their confidence limits. The minimum and maximum differences are also displayed. The t test is not significant $(t = -1.09, p = 0.2992)$, indicating that the stimuli did not significantly affect systolic blood pressure.

Note that this test of hypothesis assumes that the differences are normally distributed. This assumption can be investigated using PROC UNIVARIATE with the NORMAL option. If the assumption is not satisfied, PROC NPAR1WAY should be used.

References

Best, D.I. and Rayner, C.W. (1987), "Welch's Approximate Solution for the Behren's-Fisher Problem," *Technometrics*, 29, 205–210.

Cochran, W.G. and Cox, G.M. (1950), *Experimental Designs*, New York: John Wiley & Sons, Inc.

Freund, R.J., Littell, R.C., and Spector, P.C. (1986), *SAS System for Linear Models, 1986 Edition*, Cary, NC: SAS Institute Inc.

Huntsberger, David V. and Billingsley, Patrick P. (1989), *Elements of Statistical Inference*, Dubuque, Iowa: Wm. C. Brown Publishers.

Moore, David S. (1995), *The Basic Practice of Statistics*, New York: W. H. Freeman and Company.

Lee, A.F.S. and Gurland, J. (1975), "Size and Power of Tests for Equality of Means of Two Normal Populations with Unequal Variances," *Journal of the American Statistical Association*, 70, 933–941.

Lehmann, E. L. (1986), *Testing Statistical Hypostheses*, New York: John Wiley & Sons.

Posten, H.O., Yeh, Y.Y., and Owen, D.B. (1982), "Robustness of the Two-Sample t Test Under Violations of the Homogeneity of Variance Assumption," *Communications in Statistics*, 11, 109–126.

Ramsey, P.H. (1980), "Exact Type I Error Rates for Robustness of Student's t Test with Unequal Variances," *Journal of Educational Statistics*, 5, 337–349.

Robinson, G.K. (1976), "Properties of Student's t and of the Behrens-Fisher Solution to the Two Mean Problem," *Annals of Statistics*, 4, 963–971.

Satterthwaite, F.W. (1946), "An Approximate Distribution of Estimates of Variance Components," *Biometrics Bulletin*, 2, 110–114.

Scheffe, H. (1970), "Practical Solutions of the Behrens-Fisher Problem," *Journal of the American Statistical Association*, 65, 1501–1508.

SAS Institute Inc, (1986), *SUGI Supplemental Library User's Guide, Version 5 Edition*. Cary, NC: SAS Institute Inc.

Steel, R.G.D. and Torrie, J.H. (1980), *Principles and Procedures of Statistics*, Second Edition, New York: McGraw-Hill Book Company.

Wang, Y.Y. (1971), "Probabilities of the Type I Error of the Welch Tests for the Behren's-Fisher Problem," *Journal of the American Statistical Association*, 66, 605–608.

Yuen, K.K. (1974), "The Two-Sample Trimmed t for Unequal Population Variances," *Biometrika*, 61, 165–170.

The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/STAT® User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999.

SAS/STAT® User's Guide, Version 8

ISBN 1–58025–494–2

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

The Institute is a private company devoted to the support and further development of its software and related services.